
How are TradingView’s Exponential
Moving Average (EMA) and Relative

Moving Average (RMA) different?

TradingView’s Exponential Moving Average (EMA) and Relative Moving Average (RMA)

are calculated in almost the same way. But they do differ in one aspect. Let’s take a closer

look at how these two moving averages compare.

Formula comparison

To see why the EMA and RMA are much alike, let’s consider their formulas. That also

shows the one thing in which they differ.

TradingView calculates the Exponential Moving Average (EMA) as follows:

And for the Relative Moving Average (RMA) TradingView uses this formula:

As we can tell, both formulas are much alike. Their only difference is the alpha  variable.

Let’s see what that one is about.

‘Alpha’ value comparison

Both the EMA and RMA use an alpha  variable in their formula. This variable

determines how much weight the most recent price gets. It tells what impact the last

value has on the moving average’s entire value.

The EMA calculates that alpha  with: 2 / (length + 1) . The RMA, on the other hand,

uses 1 / length . This makes RMA’s alpha  roughly half of what the EMA uses.

To better see their difference, let’s compare alpha  values for various moving average

lengths:

Length EMA alpha Last price weight RMA alpha Last price weight

5 0.333 33.3% 0.200 20.0%

10 0.182 18.2% 0.100 10.0%

15 0.125 12.5% 0.067 6.7%

20 0.095 9.5% 0.050 5.0%

25 0.077 7.7% 0.040 4.0%

30 0.065 6.5% 0.033 3.3%

35 0.056 5.6% 0.029 2.9%

40 0.049 4.9% 0.025 2.5%

For any moving average length from 5 to 40 bars, the table above shows that EMA uses a

higher alpha  value than RMA does. So what does that mean?

Both EMA and RMA calculate their average by taking the previous bar value and

increasing that with the current bar’s value. How much of the current value the average

uses depends on alpha .

Because EMA has a higher alpha  value than RMA, the EMA gives more weight to the

last price. In fact, the weight EMA uses can be almost double of what RMA uses (see

table above).

As a result, the EMA responds faster to changes in price. Because its current value uses

more recent data than the RMA does.

Example formulas

Let’s make the above more practical. Say we have a 40-bar moving average whose

previous bar value was 26.00. The stock now closed up at 26.50. With these variables,

what value do the EMA and RMA have?

The current EMA value is 26.0256:

And the RMA is 26.0125:

These formulas show two interesting things:

For the same moving average length, the EMA gives the last price a weight of 5.1%

and uses 94.87% of the previous bar’s value. The RMA, however, only gives the latest

value a weight of 2.5% (and uses 97.5% of the old value). This makes the last price

only half as important for the RMA.

Because the EMA gives more importance to the last value, it increases faster in

response to the higher close. Both moving averages had a value of 26.00 on the

previous bar. When the stock closed higher, the EMA increased to 26.025 while the

RMA increased a little less to 26.0125.

With more weight assigned to the last value, the EMA responds faster to changes in price

direction than the RMA does. The drawback is that, when the market moves sideways,

the EMA generates more noise than the RMA.

Example script

Let’s also make a comparison with code. The indicator below plots an Exponential

Moving Average and Relative Moving Average. Both use the same data and length. We

have the code spot where the close crossed the EMA or RMA.

When both averages generate a signal at the same time, we show that confirmed signal

with a green background. If price crosses the RMA, but not the EMA, we use a blue

background. And when there’s an EMA cross, but no RMA one, we paint the bar’s

background orange. These different colours quickly show which signal happened first.

The indicator’s full code is:

//@version=5
indicator(title="RMA versus EMA", overlay=true)

// Inputs
avgData   = input.source(close, title="Average Data Source")
emaLength = input.int(20, title="EMA Length")
rmaLength = input.int(20, title="RMA Length")

// Calculate both moving averages with the same data and length
emaValue = ta.ema(avgData, emaLength)
rmaValue = ta.rma(avgData, rmaLength)

// Plot both averages on the chart
plot(rmaValue, color=color.blue, linewidth=2, title="RMA")
plot(emaValue, color=color.orange, linewidth=2, title="EMA")

// See if prices crossed the moving average
rmaCross = ta.cross(close, rmaValue)
emaCross = ta.cross(close, emaValue)

// When both averages have a cross signal on the same bar, make
// background green. Else when just one triggers, make background
// the same colour as that average's plot colour.
backgroundColour = if rmaCross and emaCross
    color.new(color.green, 75)
else if rmaCross
    color.new(color.blue, 80)
else if emaCross
    color.new(color.orange, 80)

bgcolor(backgroundColour)

We first configure the indicator with the indicator()  function. That names the script

and places it on top of the chart’s instrument.

Then we make three inputs. The first is a source input ( input.source() ). This one

configures which data the moving averages calculate on. Two integer inputs

( input.int() ) set the EMA and RMA length.

Then we calculate both moving averages:

// Calculate both moving averages with the same data and length
emaValue = ta.ema(avgData, emaLength)
rmaValue = ta.rma(avgData, rmaLength)

The ta.ema()  function calculates the Exponential Moving Average. It processes the

avgData  input variable, which defaults to close prices. Its length is the emaLength  input

variable, which has a standard value of 20 .

To get the Relative Moving Average we call the ta.rma()  function. This one also

processes the avgData  variable. Its length is rmaLength , the input variable with a period

of 20 .

With the averages computed the plot()  function shows both on the chart. The EMA

appears in blue ( color.blue ) whereas the RMA uses orange ( color.orange ).

The ta.cross()  function next looks if the bar’s close crossed over or under the RMA.

Then the function sees if such a cross happened with the EMA.

We evaluate the outcome of those crosses with a cascaded if statement. That decides

which background colour to use. First we look if both crosses happened at the same time

( and ). If that’s so, we use a transparent green ( color.green ) as the background colour.

If there wasn’t a simultaneous signal, we look if only the RMA generated a signal. If

that’s the case we pick blue as the background (the same colour as the RMA line).

Otherwise the third if  condition looks if there’s an EMA signal. If so, we use the EMA

line colour for the background (orange).

Then, on the last line of code, the bgcolor()  function applies the chosen colour to the

chart’s background.

Example chart

Here’s how the above indicator looks on the chart:

The EMA shows as an orange line while the RMA uses a blue line. We straight away see

that the EMA responds quicker than the RMA. But let’s consider the signals too.

The chart starts with a green background signal. That’s a bar where the EMA and RMA

both generated a signal at the same time.

In the chart’s middle there’s an orange background. The EMA generated a signal there

whereas the RMA did not. The next bar has a blue background: the RMA triggered a

signal (but the EMA did not). So here the RMA responded one bar later.

The right part of the chart has another orange background (EMA) first. Now it takes

three bars before the RMA generates its signal (blue background) too.

As we can tell from this, the EMA responds faster than the RMA does. Give the indicator

a try yourself and experience how different settings change the script’s signals.

Summary

The formula’s for TradingView’s Exponential Moving Average (EMA) and Relative

Moving Average (RMA) are much alike.

Their only difference is how much weight they give the most recent value. The EMA

can give that value twice as much importance than the RMA does.

That makes the EMA quicker to respond to price changes. (But, as a consequence,

also a bit more noisy.)

